DEPARTMENT OF ~~NATIONAL RESOURCES~~ NATIONAL DEVELOPMENT

BUREAU OF MINERAL RESOURCES, GEOLOGY AND GEOPHYSICS

Record 1979/28

MAG3D: AN INTERACTIVE COMPUTER PROGRAM TO CALCULATE THE MAGNETIC ANOMALY OF A FINITE DIPPING PRISM

by

R.D. Ogilvy

Record 1979/28

MAG3D: AN INTERACTIVE COMPUTER PROGRAM TO CALCULATE THE MAGNETIC ANOMALY OF A FINITE DIPPING PRISM

by

R.D. Ogilvy

CONTENTS

ILLUSTRATIONS

SUMMARY

A description is given of a computer program (MAG3D) used for calculating the magnetic anomaly due to a finite dipping prism. The program is written in FORTRAN IV for interactive use with an HP 21MX mini-computer and GOULD 5000 printer/plotter under RTE II using TEROS.

The program may be compiled, loaded, and executed using a system terminal keyboard/printer. Little or no data preparation is required and interpretation is achieved by forward interactive modelling.

The program provides a computer plot together with suitable annotation and a cross-section of the case modelled.

1. INTRODUCTION

In metalliferous exploration the need commonly arises to interpret magnetic anomalies caused by tabular bodies of finite strike length. The dipping prism model provides the interpreter with a simple yet realistic geometrical form for a first-order interpretation of an anomaly.

A computer program is described for calculating the magnetic anomaly of a finite homogeneous dipping prism. The program is called MAG3D and is based on formulae derived by Hjelt (1972). The program assumes magnetisation by induction in the Earth's field, but provision has been made to include remanent magnetisation if required.

The program has been written in FORTRAN IV for interactive use with an HP 21MX mini-computer and a GOULD 5000 printer/plotter under RTE II using TEROS. The basic program algorithm is simple and rapid, and could be readily adapted for use with some of the more modern desk top calculators. Little or no data preparation is required, and interpretation is achieved by forward interactive modelling.

2. CO-ORDINATE SYSTEM AND FORMULAE

Co-ordinate system

A rectangular co-ordinate system is used (Figs. 1a and 1b) with the X axis perpendicular to the strike direction, the Y axis parallel to the strike, and the Z axis positive downwards. The dipping surfaces of the prism are defined by the angle Φ.

The corners of the prism are defined by their X, Y, Z co-ordinates (e.g. X1, Y1, Z1) and the distances between the prism and the observation point P(X, Y, Z) are denoted by U, V, W and U',V',W'.

Oblique profiles

Profiles, other than those normal to the prism strike, can be run by defining a profile co-ordinate system, at an angle β to the prism co-ordinate system (Fig. 1e). In such cases, program MAG3D plots the calculated magnetic field with respect to the profile co-ordinate system.

Mathematical formulae

The magnetic anomaly (ΔT) of a finite dipping prism is given by Hjelt (1972):

$$\Delta T = Hx \cos I \sin \alpha + Hy \cos I \cos \alpha + Hz \sin I$$



(a) Plan

(b) Cross-section

(c) Geomagnetic field

T_i = Earths total magnetic field

I = Inclination

α = Strike angle

(d) Remanent magnetisation

T_r = Remanent component

RI = Inclination

RA = Strike angle

(e) Oblique profile

$X = xN\cos\beta - yN\sin\beta$

$Y = xN\sin\beta + yN\cos\beta$

Fig.1 Co-ordinate systems and notations

where I = inclination of the geomagnetic field; α = azimuth of prism strike (with reference to geomagnetic north)

$$\text{and } Hx = \sum_{i=1}^{3} Ji \cdot T_{1i}$$

$$Hy = \sum_{i=1}^{3} Ji \cdot T_{2i}$$

$$Hz = \sum_{i=1}^{3} Ji \cdot T_{3i}$$

and

$$J_1 = Jx.\sin\Phi - Jz.\cos\Phi$$
$$J_2 = Jy$$
$$J_3 = Jz$$

are the components of magnetisation perpendicular to the prism surfaces.

The factors T_{ij} are defined by

$$T_{11} = -\sin\Phi \cdot \theta_{1u} + \cos\Phi \cdot \theta_{2v}$$
$$T_{12} = \sin\Phi \cdot \theta_{2w}$$
$$T_{13} = \theta_{2v}$$
$$T_{21} = \theta_{2w}$$
$$T_{22} = \theta_{1v}$$
$$T_{23} = \theta_{2u}$$
$$T_{31} = \cos\Phi \cdot \theta_{1u} + \sin\Phi \cdot \theta_{2v}$$
$$T_{32} = -\cos\Phi \cdot \theta_{2w} + \theta_{2u}$$
$$T_{33} = -\theta_{1w}$$

where $\theta_{1u} = \left[\left[\left[\arctan \dfrac{V.(U.\cos\phi + W.\sin\phi)}{(U.\sin\phi - W.\cos\phi)\,.\,Ruvw}\right]_{W_2}^{W_1}\right]_{V_2}^{V_1}\right]_{U_2}^{U_1}$

$$\theta_{1v} = \left[\left[\left[\arctan \dfrac{V^2.\cos\phi - W.\,(U.\sin\phi - W.\cos\phi)}{V.\sin\phi.Ruvw}\right]_{W_2}^{W_1}\right]_{V_2}^{V_1}\right]_{U_2}^{U_1}$$

$$\theta_{1w} = \left[\left[\left[\arctan \dfrac{V\,.\,U}{W.Ruvw}\right]_{W_2}^{W_1}\right]_{V_2}^{V_1}\right]_{U_2}^{U_1}$$

$$\theta_{2u} = \left[\left[\left[\ \ln\,(U + Ruvw)\right]_{W_2}^{W_1}\right]_{V_2}^{V_1}\right]_{U_2}^{U_1}$$

$$\theta_{2v} = \left[\left[\left[\ \ln\,(V + Ruvw)\right]_{W_2}^{W_1}\right]_{V_2}^{V_1}\right]_{U_2}^{U_1}$$

$$\theta_{2w} = \left[\left[\left[\ \ln\,(U.\cos\phi + W\,.\,\sin\phi + Ruvw)\right]_{W_2}^{W_1}\right]_{V_2}^{V_1}\right]_{U_2}^{U_1}$$

and $Ruvw = (u^2 + v^2 + w^2)^{1/2}$

If the prism also has remanent magnetisation, the total magnetisationof the prism is given by

$Jx = K\ (Tix + Trx)$

$Jy = K\ (Tiy + Try)$

$Jz = K\ (Tiz + Trz)$

where the subscripts ix, rx, etc. refer to the induced and remanent components respectively, in the appropriate X, Y, or Z directions.

These values may be substituted directly in the above sets of equations.

3. PROGRAM DESCRIPTION

Flow chart

The essential elements of the program structure are shown in the flow chart (Fig. 2). Program variables used in the flow chart are defined in the program listing (see Appendix). For operational convenience in BMR, a number of system subroutines are used, but these do not detract from the generality of the basic program algorithm. System routines are referred to by name (e.g. RMPAR, NUMB, etc.) and further details may be obtained from the HP-RTE II system manual and the GOULD printer manual.

Data input

The input model parameters used in the program are defined as follows:

MN	=	Model number. Always start with MN = 1
X1	=	X co-ordinate of top edge nearest origin
X2	=	X co-ordinate of top edge farthest from origin

Y1	=	Y co-ordinate of top edge nearest origin
Y2	=	Y co-ordinate of top edge farthest from origin
Z1	=	Depth to top of prism
Z2	=	Depth to base of prism
XN	=	X co-ordinate of first observation point
YN	=	Y " " " " "
ZN	=	Z " " " " "
DX	=	Station increment along axis
NXN	=	Number of stations
PHI	=	Dip of prism edge (degrees)
ALPHA	=	Angle between prism strike and magnetic north (degrees)
BETA	=	Angle between X axis and profile, measured clockwise from the positive X axis (degrees)
SUSC	=	Susceptibility contrast
TOTAL	=	Earth's total field
DIP	=	Inclination of Earth's field
QKR	=	Koenigsberger ratio
RA	=	Angle between magnetic north and remanent vector
RI	=	Inclination of remanent vector with horizontal

The program is structured for maximum interactive use by the interpreter, and may be compiled, loaded, and executed by means of a terminal keyboard/printer under TEROS. Model parameters for each case are entered in free-field format.



Fig. 2 Program flow chart

The source program is stored as a resident disc file MAG3DS but is also available on paper tape.

Loading the program

To load the program from source file, the following (TEROS) command sequence is required:

1.	*GF, MAG3DS	This command calls the source program into the user scratch region.
2.	*FT, FL	This command causes the program to be compiled as a binary relocatable file.
3.	*GF, STAT..	This command schedules the binary file for loading.
4.	*LD, 0	This command causes the program to be loaded.

The above steps are once-only commands and need not be repeated while the system is up.

Running the program

To run the program, the following command sequence is observed:

1.	*LG, MAG3D	This command causes scheduling of the program for execution.
2.	"MN, X1, X2, Y1, Y2, Z1, Z2 ="	At this request, type in the appropriate model parameters.
3.	"XN, YN, ZN, DX, NXN, TOTAL, DIP, SUSC, PHI, ALPHA, BETA="	Type in appropriate parameters separated by a comma.
4.	"REMANENT MAGNETISATION?"	Answer YES or NO. If NO, the program will skip to step 6.
5.	"QKR, RI, RA="	Type in the appropriate parameters.
6.	"MISTAKE IN INPUT DATA?"	Answer YES or NO. If YES, the program will return to step 2.
7.	"ANY MORE CASES?"	Answer YES or NO. If YES, the program will return to step 2 so that additional cases can be modelled.

The user may log into the program any number of times using the command *LG, MAG3D. User commands are preceded by a * as shown; terminal replies are given in inverted commas.

An example, illustrating the use of the input commands is given in Figure 3.

If more than one case is to be modelled, all angular parameters MUST be redefined; other parameters can be redefined as required. For all parameters preceding those to be changed, type in commas. Following parameters may be ignored. All parameters must be initially defined.

Data output

Two modes of data output are utilised. The calculated magnetic values are sent to the user teletype for immediate perusal (Fig. 3) and also to the line printer for plotting. The graphical plot includes suitable annotation and a cross-section of the case modelled (see Fig. 4). A cross-section of the model is not produced for oblique profiles.

4. PROGRAM TESTING

The program was tested against Bhattacharyya's (1964) formula for the magnetic effect of a right rectangular prism, and against Gay's (1963) formula for the magnetic effect of a two-dimensional inclined dyke.

Comparison with Gay's model was obtained by setting V2 = V1 and taking a near-infinite strike length (i.e. V1 = -V2 = 1000 m). The results shown in Figure 4 indicate a very close fit between the test data and the analytical solutions of Bhattacharyya and Gay.

5. REFERENCES

BHATTACHARYYA, B.K., 1964 - Magnetic anomalies due to prism-shaped bodies with arbitrary polarization. Geophysics, v. 29, pp. 517-531.

HJELT, S.-E., 1972 - Magnetostatic anomalies of dipping prisms. Geoexploration, v. 10(4), pp. 239-254.

GAY, S.P., Jr, 1963 - Standard curves for interpretation of magnetic anomalies over long tabular bodies. Geophysics, v. 28, pp. 161-200.

Loading program

```
◆GF,MAG3DS
MAG3DS COPIED TO SCRATCH TRACKS,  LENGTH; 274 RECORDS, 3700 WORDS
#◆FT,FL
TEROS TERMINAL ASSEMBLY/COMPILATION SYSTEM. BINARY OUTFILE =  STAT1D
#◆GF,STAT1D
STAT1D COPIED TO SCRATCH TRACKS,  LENGTH; 108 RECORDS, 3700 WORDS
#◆LD,0
```

```
 /LOADR:LOAD
 /LOADR:MAG3D-READY
 /LOADR:$END
```

Running program

```
#◆LG,MAG3D
MAG3D ON

  MN,X1,X2,Y1,Y2,Z1,Z2=
1,60.,100.,20.,60.,20.,60.

  XN,YN,ZN,DX,NXN,TOTAL,DIP,SUSC,PHI,ALPHA,BETA=
0.,40.,0.,10.,20,58000.,-45.,0.005,90.,80.,0.

  REMANENT MAGNETIZATION?
NO

  MISTAKE IN INPUT DATA?
NO


1         MODEL NO.           1
0         STATION            TOTAL FIELD
0              .00           -18.53
0            10.00           -28.50
0            20.00           -44.47
0            30.00           -69.66
0            40.00          -106.21
0            50.00          -143.36
0            60.00          -131.27
0            70.00           -25.60
0            80.00           122.79
0            90.00           255.28
0           100.00           319.19
0           110.00           281.13
0           120.00           201.87
0           130.00           135.50
0           140.00            90.41
0           150.00            61.27
0           160.00            42.47
0           170.00            30.15
0           180.00            21.90
0           190.00            16.26

  ANY MORE CASES?
YES

  MN,X1,X2,Y1,Y2,Z1,Z2=

       ETC.ETC.
```

Fig.3 Teletype input/output





Fig.4 Program testing

Appendix: PROGRAM LISTING

```
****************************************************************************************************
**
**
**
**
**
**
**
**                              MAG3D
**                          oTERMINAL      28
**                    OUTPUT LOGICAL UNIT =       6
**
**
**
**
**
**
**
****************************************************************************************************
```

```
FTN4,B,L
      PROGRAM MAG3D
C AUTHOR- R.D.OGILVY
C THIS PROGRAM CALCULATES THE MAGNETIC ANOMALY OF A DIPPING PRISM
C MN=MODEL NO.
C X1=COOD OF FIRST TOP EDGE ALONG X AXIS
C X2=COOD OF SECOND TOP EDGE ALONG X AXIS
C
C Y1=COOD OF FIRST EDGE ON Y AXIS
C Y2=COOD OF SECOND EDGE ON Y AXIS
C Z1=DEPTH TO TOP OF PRISM
C Z2=DEPTH TO BASE OF PRISM
C XN=X COOD OF FIRST FIELD POINT
C YN=Y COOD OF FIRST FIELD POINT
C ZN=Z COOD OF FIRST FIELD POINT
C DX=STATION INCREMENT ALONG PROFILE
C PHI=DIP OF PRISM (MEASURED CLOCKWISE FROM +VE X AXIS)
C DIP=INCLINATION OF EARTHS FIELD (DEGREES),  -VE IN SOUTHERN HEMISPHERE.
C TOTAL=EARTHS TOTAL FIELD(NT)
C SUSC=SUSCEPTIBILITY(C.G.S)
C ALPHA=ANGLE BETWEEN PRISM STRIKE AND MAGNETIC NORTH
C     *MEASURED CLOCKWISE FROM MN TO +VE Y AXIS.
C BETA=ANGLE BETWEEN X AXIS AND PROFILE,MEASURED CLOCKWISE FROM
C     +VE X AXIS.
C NXN=NUMBER OF STATIONS
C AT=ARRAY FOR TOTAL MAGNETIC FIELD VALUES
C STAT=ARRAY FOR OBSERVATION POINTS
C
C IREM=NO (REMANENCE ABSENT)
C IREM=YES(REMANENCE PRESENT)
C RI=INCLINATION OF REMANENT VECTOR WITH HORIZONTAL
C RA=ANGLE BETWEEN MAGNETIC NORTH AND REMANENT VECTOR
C QKR=KONIGSBERGER RATIO
C
      DIMENSION F(8),U(2),V(2),W(2),G(8),B(8),S(8)
      DIMENSION T(8),A(8),AT(100),STAT(100),LU(5)
      INTEGER STR1(8),STR2(7),STR3(9)
      INTEGER STR4(8),STR5(8),STR6(8),STR7(8),STR8(8),STR9(8)
      INTEGER STR10(8),STR11(7),STR12(6),STR13(7)
      DATA STR1/2HTO,2HTA,2HL ,2HFI,2HEL,2HD(,2HNT,2H) /
      DATA STR2/2HDE,2HPT,2HH(,2HME,2HTR,2HES,2H) /
      DATA STR3/2HDI,2HST,2HAN,2HCE,2H (,2HME,2HTR,2HES,2H) /
      DATA STR4/2HMO,2HDE,2HL ,2HNO,2H. ,2H  ,2H  ,2H   /
      DATA STR5/2HIN,2HDU,2HCI,2HNG,2H F,2HIE,2HLD,2H   /
      DATA STR6/2HIN,2HCL,2HIN,2HAT,2HIO,2HN ,2H  ,2H   /
      DATA STR7/2HSU,2HSC,2HEP,2HTI,2HBI,2HLI,2HTY,2H   /
      DATA STR8/2HPR,2HIS,2HM ,2HDE,2HPT,2HH ,2H  ,2H   /
      DATA STR9/2HPR,2HIS,2HM ,2HDI,2HP ,2H  ,2H  ,2H   /
      DATA STR10/2HPR,2HIS,2HM ,2HST,2HRI,2HKE,2H  ,2H   /
      DATA STR11/2HCR,2HOS,2HS-,2HSE,2HCT,2HIO,2HN /
      DATA STR12/2HPR,2HOF,2HIL,2HE ,2H(Y,2HN)/
      DATA STR13/2HPR,2HOF,2HIL,2HE ,2HAN,2HGL,2HE /
C READ INPUT DATA
C
```

```
      CALL RMPAR(LU)
    1 WRITE(LU,300)
  300 FORMAT(/2X,"MN,X1,X2,Y1,Y2,Z1,Z2=")
      READ(LU,*)MN,X1,X2,Y1,Y2,Z1,Z2
      IF(MN.LT.0)GO TO 27
      WRITE(LU,301)
  301 FORMAT(/2X,"XN,YN,ZN,DX,NXN,TOTAL,DIP,SUSC,PHI,ALPHA,BETA=")
      READ(LU,*)XN,YN,ZN,DX,NXN,TOTAL,DIP,SUSC,PHI,ALPHA,BETA
      WRITE(LU,305)
  305 FORMAT(/2X,"REMANENT MAGNETIZATION?")
      READ(LU,17)IREM
   17 FORMAT(A2)
      IF(IREM.NE.2HYE)GO TO 14
   13 WRITE(LU,302)
  302 FORMAT(/2X,"QKR,RI,RA=")
      READ(LU,*)QKR,RI,RA
   14 WRITE(LU,303)
  303 FORMAT(/2X,"MISTAKE IN INPUT DATA?")
      READ(LU,5)NGOOD
    5 FORMAT(A2)
      IF(NGOOD.EQ.2HYE)GO TO 1
C
   29 PI=3.14159
      PHI=PHI*PI/180.
      DIP=DIP*PI/180.
      ALPHA=ALPHA*PI/180.
      RI=RI*PI/180.
      RA=RA*PI/180.
      BETA=BETA*PI/180.
      XM=XN
C
      DO 99 L=1,NXN
       X=(L-1)*DX*COS(BETA)+XN
       Y=(L-1)*DX*SIN(BETA)+YN
       STAT(L)=XN+(L-1)*DX
      H=Z2-Z1
      U(1)=X-X2
      U(2)=X-X1
      V(1)=Y-Y2
      V(2)=Y-Y1
      W(1)=ZN-Z2
      W(2)=ZN-Z1
C
C CALCULATE F1U ,F1V,F1W,F2U,F2V,F2W
C
      M=1
      F1U=0.
      F1V=0.
      F1W=0.
      F2U=0.
      F2V=0.
      F2W=0.
      DO 10 I=1,2
      DO 10 J=1,2
      DO 10 K=1,2
```

```
      IF(K.EQ.1)U(I)=U(I)-H/TAN(PHI)
      IF((ABS(U(I))-0.001).LT.0)U(I)=0.001
      IF((ABS(V(J))-0.001).LT.0)V(J)=0.001
      IF((ABS(W(K))-0.001).LT.0)W(K)=0.001
      R=SQRT(U(I)*U(I)+V(J)*V(J)+W(K)*W(K))
      P=U(I)*COS(PHI)+W(K)*SIN(PHI)
      Q=U(I)*SIN(PHI)-W(K)*COS(PHI)
      IF((ABS(Q)-0.001).LT.0)Q=0.001
      F(M)=ATAN((V(J)*P)/(Q*R))
      G(M)=ATAN((V(J)*V(J)*COS(PHI)-W(K)*Q)/(V(J)*SIN(PHI)*R))
      B(M)=ATAN((V(J)*U(I))/(W(K)*R))
      S(M)=ALOG(U(I)+R)
      T(M)=ALOG(V(J)+R)
      A(M)=ALOG(P+R)
      IJK=I+J+K
      IF(IJK.EQ.4.OR.IJK.EQ.6)GO TO 7
    6 F(M)=-F(M)
      G(M)=-G(M)
      B(M)=-B(M)
      S(M)=-S(M)
      T(M)=-T(M)
      A(M)=-A(M)
    7 IF(K.EQ.1)U(I)=U(I)+H*COS(PHI)/SIN(PHI)
      M=M+1
   10 CONTINUE
      DO 11 N =1,8
      F1U=F1U+F(N)
      F1V=F1V+G(N)
      F1W=F1W+B(N)
      F2U=F2U+S(N)
      F2V=F2V+T(N)
       F2W=F2W+A(N)
   11 CONTINUE
C
C CALCULATE T FACTORS
C
      T11=COS(PHI)*F2V-SIN(PHI)*F1U
      T12=SIN(PHI)*F2W
      T13=F2V
      T21=F2W
      T22=F1V
      T23=F2U
      T31=COS(PHI)*F1U+SIN(PHI)*F2V
      T32=F2U-COS(PHI)*F2W
      T33=-F1W
C
C CALCULATION OF H FACTORS
C
      TOX=TOTAL*COS(DIP)*SIN(ALPHA)
      TOY=TOTAL*COS(DIP)*COS(ALPHA)
      TOZ=TOTAL*SIN(DIP)
      IF(IREM.EQ.2HYE)GO TO 613
      AJX=SUSC*TOX
      AJY=SUSC*TOY
      AJZ=SUSC*TOZ
```

```
  614  AJ1=AJX*SIN(PHI)-AJZ*COS(PHI)
       AJ2=AJY
       AJ3=AJZ
       HX=AJ1*T11+AJ2*T12+AJ3*T13
       HY=AJ1*T21+AJ2*T22+AJ3*T23
        HZ=AJ1*T31+AJ2*T32+AJ3*T33
C
C CALCULATION OF TOTAL FIELD
C
        AT(L)=HX*COS(DIP)*SIN(ALPHA)+HY*COS(DIP)*COS(ALPHA)+HZ*SIN(DIP)
   99 CONTINUE
       XN=XM
       GO TO 550
C CALCULATION OF REMANENT MAGNETIZATION
C
  613  TR=QKR*TOTAL
       TRX=TR*COS(RI)*SIN(ALPHA-RA)
       TRY=TR*COS(RI)*COS(ALPHA-RA)
       TRZ=TR*SIN(RI)
       AJX=SUSC*(TOX+TRX)
       AJY=SUSC*(TOY+TRY)
       AJZ=SUSC*(TOZ+TRZ)
       GO TO 614
C
C PRINT RESULTS
C
  550 WRITE(LU,3)MN
       WRITE(6,3)MN
    3 FORMAT(1H1,7X,9HMODEL NO.,I10)
       WRITE(LU,202)
       WRITE(6,202)
  202 FORMAT(1H0,7X,7HSTATION,10X,11HTOTAL FIELD)
       DO 56 J=1,NXN
       WRITE(6,60)STAT(J),AT(J)
       WRITE(LU,60)STAT(J),AT(J)
   60 FORMAT(1H0,5X,F10.2,5X,F10.2)
   56 CONTINUE
C
C        PLOT RESULTS
C
       SV1=TOTAL-200.
       SI1=200.00
       SV2=Z2
       SI2=-Z2/3.
       DIST=DX*FLOAT(NXN)/10.
       CALL PLOTS(17.,10.,0)
       CALL FACTR(0.5)
       IF(MN.EQ.1)CALL PLOT(1.,0.5,-3)
       CALL AXIS(0.,4.,STR1,16,5.,90.,SV1,SI1)
       CALL AXIS(0.,0.,STR2,14,3.,90.,SV2,SI2)
       CALL AXIS(0.,3.,STR3,18,10.,0.,XM,DIST)
C
C        PLOT  POINTS
C
       CALL PLOT(0.,5.,-3)
```

```
      XP=0
      DO 600 I=1,NXN
      YP=AT(I)/200.
      CALL SYMB(XP,YP,0.1,7,0.,-1,0.1)
      XP=XP+DX/DIST
  600 CONTINUE
      BETA=BETA*180./PI
      IF(IFIX(BETA).GT.0)GO TO 900
      CALL PLOT(0.,-2.,-3)
      XP1=X1/DIST-XM/DIST
      YP1=Z1/SIZ
      XP2=X2/DIST-XM/DIST
      XP3=(X2+H/TAN(PHI))/DIST-XM/DIST
      XP4=(X1+H/TAN(PHI))/DIST-XM/DIST
      CALL PLOT(XP1,YP1,3)
      CALL PLOT(XP2,YP1,2)
      CALL PLOT(XP3,-3.,2)
      CALL PLOT(XP4,-3.,2)
      CALL PLOT(XP1,YP1,2)
      CALL SYMB(7.,-1.5,0.1,STR11,0.,13)
  900 DIP=DIP*180./PI
      PHI=PHI*180./PI
      ALPHA=ALPHA*180./PI
      CALL SYMB(7.,5.5,0.1,STR4,0.,16)
      CALL SYMB(7.,5.,0.1,STR5,0.,16)
      CALL SYMB(7.,4.7,0.1,STR6,0.,16)
      CALL SYMB(7.,4.4,0.1,STR7,0.,16)
      CALL SYMB(7.,4.1,0.1,STR8,0.,16)
      CALL SYMB(7.,3.8,0.1,STR9,0.,16)
      CALL SYMB(7.,3.5,0.1,STR10,0.,16)
      CALL SYMB(7.,3.2,0.1,STR12,0.,12)
      CALL SYMB(7.,2.9,0.1,STR13,0.,14)
      CALL NUMB(8.8,5.5,0.1,FLOAT(MN),0.,0)
      CALL NUMB(8.8,5.0,0.1,TOTAL,0.,2)
      CALL NUMB(8.8,4.7,0.1,DIP,0.,2)
      CALL NUMB(8.8,4.4,0.1,SUSC,0.,6)
      CALL NUMB(8.8,4.1,0.1,Z1,0.,2)
      CALL NUMB(8.8,3.8,0.1,PHI,0.,2)
      CALL NUMB(8.8,3.5,0.1,ALPHA,0.,2)
      CALL NUMB(8.8,3.2,0.1,YN,0.,2)
      CALL NUMB(8.8,2.9,0.1,BETA,0.,2)
C
C     TERMINATE PLOT
C
   27 CALL PLOT(0.,0.,999)
      CALL GOPLT
      WRITE(LU,80)
   80 FORMAT(/2X,"ANY MORE CASES?")
      READ(LU,81)MORE
   81 FORMAT(A2)
      IF(MORE.EQ.2HYE)GO TO 1
      STOP
      END
      END$
```